



Ethos Audit Report

Version 1.0

056Security

December 11, 2024

Protocol Audit Report

056Security

December, 2024

Prepared by: 056Security

Lead Auditors:

- 0xb0k0
- stanchev

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - Medium
 - Low
 - Informational

Protocol Summary

[Ethos](#) in its essence is a platform that strives to solve the problem of “social validation” and allow users to properly validate their identity and the identity of others. Users create their profiles, review other profiles and vouch for them. All of these actions increase their online reputation score and help them get more verified. The platform also tries to build a sort of a reputation market, where people can buy and sell votes based on the reputation score. [056Security](#) was hired from the [Ethos](#) team to help them organize and prepare the financial part of the protocol (i.e [EthosVault](#) (which was removed after as it had critical issues), [EthosVouch](#), [ReputationMarket](#)) for their upcoming contest on one of the leading platforms. The whole process took about ~1 month.

Disclaimer

056Security makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts. We highly recommend that additional audits be done to ensure the security of the protocol.

Risk Classification

		Critical	High	Medium	Low
Likelihood	Critical	C	C/H	H	M
	High	C/H	H	H/M	M
	Medium	H	H/M	M	M/L
	Low	M	M	M/L	L

Audit Details

Scope

- Commit Hash: Various commits reaching [215e22a78da37631812acf929b45cd3726a3e374](#)

- In Scope:

```
1 ./contracts/  
2 - EthosVault.sol  
3 - EthosVaultManager.sol  
4 - EthosFees.sol  
5 - EthosSlashPenalty.sol  
6 - EthosVouch.sol  
7 - ReputationMarket.sol
```

- Solc Version: 0.8.26
- Chain(s) to deploy contract to: Base
- Tokens:
 - ETH

Executive Summary

The audit resulted in **four critical**, **two high**, **four medium**, **two low**, **four gas** and **two informational** findings.

Issues found

Severity	Number of issues found
Critical	4
High	2
Medium	4
Low	2
Informational	2
Gas	4
Total	18

Findings

Critical

[C-01] User can deposit directly into the vault without providing ETH

Description: Currently, the vault's `deposit()` function is not set as `payable`, which means that the user can deposit assets to the vault without spending actual funds. When the internal `_deposit()` function is called, it will take the `WETH` deposit amount from the vault's balance, instead of the user's balance. What is more, the `msg.sender` will get minted shares, which will be redeemable. A malicious user can quickly drain the vault by depositing assets and then calling `withdraw()` to get the funds back.

```
1  function deposit(uint256 assets, address receiver) public virtual
    override returns (uint256) {
2      uint256 stakes = previewDeposit(assets);
3      _deposit(msg.sender, receiver, assets, stakes);
4      return stakes;
5  }
6
7  function _deposit(
8      address /caller/,
9      address receiver,
10     uint256 assets,
11     uint256 stakes
12 ) internal virtual {
13     super._mint(receiver, stakes);
14     IWETH9(asset()).deposit{ value: assets }();
15     emit DepositETH(msg.sender, assets, stakes);
16     _updateSlashPoints(stakes, receiver);
17 }
```

Impact: Vault funds will be drained.

Mitigation: Make the `deposit()` function `payable` and check if the user has provided the correct amount of ETH.

[C-02] Profile with multiple addresses that vouches may end up with stuck funds for addresses that did not initiate the vouch

Description: If we have a profile that represents an organization with multiple addresses, a problem could arise with vouching where after one address unvouches, the other addresses that also vouch for the same profile will have their shares stuck, as the vouch won't be redeemable anymore.

Impact: Stuck funds for users.

Proof of concept:

Organization Ethos&056 has 3 addresses registered to the profile (Ben,Boris,Ivan)

1. Ben vouches for Alice with 1 eth(vault is created shares are distributed)
2. Boris increases the vouch for Alice with 1 eth(vault is created shares are distributed)
3. Ivan increases the vouch for Alice with 0.1 eth
4. Ivan decides to unvouch as he doesn't like Alice anymore
5. Ivan gets his 0.1eth back pays the fee
6. Now Ben and Boris have stuck funds as the vouch is no longer redeemable(archived)

Recommended mitigation: Keep track of all vouches for a profile and make sure that whenever there is an unvouch, it is applied per address.

[C-03] Malicious user could use the EthosVault::withdraw() function to drain the vault contract

Description: Due to improper checks in the `_withdraw()` function, a malicious user can drain the vault contract by withdrawing other users funds. Currently, the `_withdraw()` function does no checks if the `msg.sender` is actually the owner of the shares. This leads to the `msg.sender` being able to withdraw other users funds.

```
1 function withdraw(  
2     uint256 assets,  
3     address receiver,  
4     address owner  
5 ) public virtual override returns (uint256) {  
6     uint256 stakes = previewWithdraw(assets);  
7     _withdraw(msg.sender, receiver, owner, assets, stakes);  
8     emit WithdrawETH(receiver, assets, stakes);  
9     return stakes;  
10 }  
11  
12 function _withdraw(  
13     address caller, // vouch contract  
14     address receiver, // vouch contract  
15     address owner, // voucher  
16     uint256 assets, // amount of ETH to withdraw  
17     uint256 stakes // amount of stakes to burn  
18 ) internal virtual returns (uint256) {  
19     @> require(caller == receiver, "Only the vouch contract can withdraw  
20         assets"); // @audit - both caller and receiver are inputs to the  
21         withdraw function  
22     // the owner holds N stakes of M assets  
23     // in order to "give back" their stakes, we burn them  
24     super._burn(owner, stakes);
```

```
23     // in exchange, we retrieve the assets
24     // first by converting from WETH to ETH
25     IWETH9(asset()).withdraw(assets);
26     // then sending the ETH to the receiver (in this case, the Vouch
        contract, so we can apply fees)
27     (bool success, ) = payable(receiver).call{ value: assets }("");
28     require(success, "Failed to send ETH");
29     emit WithdrawETH(owner, assets, stakes);
30     return assets;
31 }
```

Impact: Vault funds will be drained.

Mitigation: Add checks to make sure that the `msg.sender` is actually the owner of the shares or make the `withdraw()` function only callable by the vouch contract.

[C-04] Vault provides a `donate()` function which implies for the classical donation attack

Description: The `donate()` function is a function that allows users to donate ETH directly to the vault. This opens up the way for a classical donation attack, where a malicious user can deposit `wei` amounts and directly donate a large amount of ETH, making every other deposit worthless, as it will not mint any shares. This allows the donator to withdraw his/her `wei` amount of share, which will be worth all the ETH in the vault.

Impact: Users will not get any shares when depositing, and their actual funds will be drained.

Mitigation: Remove the `donate()` function and potentially re-think the design of the `EthosVault`.

High

[H-01] Math rounding in `EthosVault` is not ERC4626-compliant: `previewWithdraw` should round ceil

Description: Quote from <https://eips.ethereum.org/EIPS/eip-4626#security-considerations>:

Finally, EIP-4626 Vault implementers should be aware of the need for specific, opposing rounding directions across the different mutable and view methods, as it is considered most secure to favor the Vault itself during calculations over its users: (1) it's calculating how many shares to issue to a user for a certain amount of the underlying tokens they provide or (2) it's determining the amount of the underlying tokens to transfer to them for returning a certain amount of shares, it should round down. If (1) it's calculating the amount of shares a user has to supply to receive a given

amount of the underlying tokens or (2) it's calculating the amount of underlying tokens a user has to provide to receive a certain amount of shares, it should round up.

This could be a little confusing, but in order to understand it fully you can open OZ's [erc4626](#) implementation. There they have all the `previewWithdraw`, `previewDeposit` implementation and where they should round to.

Impact: Users will not get the correct amount of shares when withdrawing.

Mitigation: Add the following changes to the `previewWithdraw` function:

```
1 function previewWithdraw(uint256 assets) public view virtual override
  returns (uint256) {
2     return
3         totalSupply() == 0
4         ? assets
5         : Math.mulDiv(
6             assets, // New assets being added
7             totalSupply(), // Current total stakes
8             totalAssets(), // Current total assets
9             Math.Rounding.Ceil
10        );
11 }
```

Or simplify the vault to look exactly like [erc4626](#) OZ in order to have proper structure `preview` -> `convertToAssets/Shares`

[H-02] EthosVault is missing slippage protection on it's functions

Description: The current implementation of [ERC4626](#) does not include slippage protection mechanisms for deposit, mint, withdraw, and redeem operations as recommended by [EIP-4626](#). This omission could lead to unexpected losses for users, especially EOAs interacting directly with the contract.

If implementors intend to support EOA account access directly, they should consider adding an additional function call for deposit/mint/withdraw/redeem with the means to accommodate slippage loss or unexpected deposit/withdrawal limits, since they have no other means to revert the transaction if the exact output amount is not achieved.

Impact: Users will not be able to withdraw the exact amount of ETH they want.

Mitigation: Add slippage protection in the following functions:

```
1 function depositWithSlippage(uint256 assets, address receiver, uint256
  minSharesOut) public returns (uint256 shares);
```



```
2 function mintWithSlippage(uint256 shares, address receiver, uint256
    maxAssetsIn) public returns (uint256 assets);
3 function withdrawWithSlippage(uint256 assets, address receiver, address
    owner, uint256 maxSharesBurned) public returns (uint256 shares);
4 function redeemWithSlippage(uint256 shares, address receiver, address
    owner, uint256 minAssetsOut) public returns (uint256 assets);
```

Medium

[M-01] The slippage check in ReputationMarket::buyVotes round incorrectly in certain cases

Description: The current logic for slippage checks in ReputationMarket::buyVotes does not work as expected. In certain cases, as when a user sets a slippage of 0.001%, if he/she wants to buy 8 votes, but due to the market conditions, he/she gets 7 votes, the transaction will revert, even though the slippage tolerance is extremely low. Due to Solidity's rounding down, even the smallest possible slippage will be rounded down.

```
1 /**
2  * @dev Checks if the actual votes bought are within range of the
3  *   expected votes bought given a slippage tolerance in basis points.
4  * @param actual The actual votes bought.
5  * @param expected The expected votes bought.
6  * @param slippageBasisPoints The slippage tolerance in basis points
7  *   (1 basis point = 0.01%).
8  */
9 function _checkSlippageLimit(
10     uint256 actual,
11     uint256 expected,
12     uint256 slippageBasisPoints
13 ) private pure {
14     uint256 toleratedVotes = (expected * ((SLIPPAGE_POINTS_BASE -
15         slippageBasisPoints))) /
16         SLIPPAGE_POINTS_BASE;
17     if (actual < toleratedVotes) {
18         revert SlippageLimitExceeded(actual, expected,
19             slippageBasisPoints);
20     }
21 }
```

Impact: Reverted transactions due to incorrect slippage checks.

Recommended mitigation: Use special Math libraries for proper floor/ceil rounding.

[M-02] Using `transfer` and `send` for ETH transfers in Reputation Market

Description: When transferring ETH to recipients, if Solidity's `transfer()` or `send()` method is used, certain shortcomings arise, particularly when the recipient is a smart contract. These shortcomings can make it impossible to successfully transfer ETH to the smart contract recipient.

The following cases could lead to failed transactions:

1. The recipient does not implement a payable fallback function.
2. The recipient implements a payable fallback function which would incur more than 2300 gas units.
3. The recipient implements a payable fallback function incurring less than 2300 gas units but is called through a proxy that raises the call's gas usage above 2300.

Impact: Reverted transactions.

Recommended mitigation: Use `call` with its returned boolean checked in combination with re-entrancy guard.

[M-03] There is no way of withdrawing any dust ether

Description: Since the Vault can receive native ETH, it is inevitable that a surplus may accumulate and remain stuck in the contract, as there is no way to withdraw it.

Impact: Stuck funds.

Mitigation: Add a proper way to withdraw stuck funds in the contract

[M-04] Chance of reentrancy in Ethos protocol

Description: The code contains multiple low level calls that could lead to reentrancy vulnerabilities.

Impact: Malicious user could reenter the contract and vouch maliciously.

Recommended mitigation: Add `NonReentrant` modifier whenever there is a low level call.

Low**[L-01] `EthosVouch::updateMaximumVouches` uses `>=` instead of `>`**

Description: The default value for `maximumVouches` is 256, so if the admin changes it to a different value, it won't be able to set it to 256 again.

```
1 function updateMaximumVouches(uint256 maximumVouches_) external
  onlyAdmin whenNotPaused {
2   @> if (maximumVouches_ >= 256) { // @audit - should be > so that it
    can be set to 256
3     revert MaximumVouchesExceeded(maximumVouches_, "Maximum vouches
      cannot exceed 256");
4   }
5   maximumVouches = maximumVouches_;
6 }
```

Recommended mitigation: Use > instead of >=.

[L-02] Remove `mint()` as it is not used anywhere in the code

Description: While it is public and could be called by anyone without giving any weth,eth at stake. It will probably revert because `_deposit` will wrap 0 eth which is impossible.

Mitigation: Remove `mint()`.

Informational

[INFO-01] Use shares instead of stakes for more understandable naming

Mitigation: Change naming of stakes to shares everywhere as it is more understandable

[INFO-02] No need to apply modifier checks in the `EthosVaultManager::initialize()` function as `__accessControl_init` checks them

Description: The modifiers in the `EthosVaultManager::initialize()` function are redundant as `__accessControl_init` checks them.

```
1 function initialize(
2   address _owner,
3   address _admin,
4   address _expectedSigner,
5   address _signatureVerifier,
6   address _contractAddressManagerAddr
7 )
8   public
9   initializer
10  validAddress(_owner)
11  validAddress(_admin)
12  validAddress(_expectedSigner)
13  validAddress(_signatureVerifier)
```

```
14     validAddress(_contractAddressManagerAddr)
15     {...}
```

Gas

[G-01] EthosVouch::markUnhealthy(...) has a duplicated IEthosProfile(...).verifiedProfileIdForAddress(...) call

The `markUnhealthy` function makes a call to `IEthosProfile(...).verifiedProfileIdForAddress(...)` twice, where the second call does utilize the returned data.

```
1     function markUnhealthy(uint256 vouchId) public whenNotPaused {
2         Vouch storage v = vouches[vouchId];
3         uint256 profileId = IEthosProfile(
4             contractAddressManager.getContractAddressForName(ETHOS_PROFILE)
5             ).verifiedProfileIdForAddress(msg.sender);
6
7         _vouchShouldExist(vouchId);
8         _vouchShouldBePossibleUnhealthy(vouchId);
9         _vouchShouldBelongToAuthor(vouchId, profileId);
10    @> IEthosProfile(contractAddressManager.getContractAddressForName(
11        ETHOS_PROFILE))
12        .verifiedProfileIdForAddress(msg.sender); // @audit - duplicated
13        redundant call
14
15        v.unhealthy = true;
16        // solhint-disable-next-line not-rely-on-time
17        v.activityCheckpoints.unhealthyAt = block.timestamp;
18
19        emit MarkedUnhealthy(v.vouchId, v.authorProfileId, v.
20            subjectProfileId);
21    }
```

[G-02] Store storage variables in local variables when used multiple times in a function

```
1     function _rewardPreviousVouchers(
2         uint256 amount,
3         uint256 subjectProfileId
4     ) internal returns (uint256 amountDistributed) {
5         uint256[] storage vouchIds = vouchIdsForSubjectProfileId[
6             subjectProfileId];
7
8         // Calculate total balance of all active vouches
9         uint256 totalBalance;
10    @> for (uint256 i = 0; i < vouchIds.length; i++) { // @audit - store
11        vouchIds.length in a local variable
```

```
10     Vouch storage vouch = vouches[vouchIds[i]];
11     // Only include active (not archived) vouches in the distribution
12     if (!vouch.archived) {
13         totalBalance += vouch.balance;
14     }
15 }
16
17 // If this is the first voucher, do not distribute rewards
18 if (totalBalance == 0) {
19     return totalBalance;
20 }
21
22 // Distribute rewards proportionally
23 uint256 remainingRewards = amount;
24 for (uint256 i = 0; i < vouchIds.length && remainingRewards > 0; i++) {
25     Vouch storage vouch = vouches[vouchIds[i]];
26     if (!vouch.archived) {
27         // Calculate this vouch's share of the rewards
28         uint256 reward = amount.mulDiv(vouch.balance, totalBalance,
29             Math.Rounding.Floor);
30         if (reward > 0) {
31             vouch.balance += reward;
32             remainingRewards -= reward;
33         }
34     }
35 }
36 // Send any dust (remaining rewards due to rounding) to the subject
37 // reward escrow
38 if (remainingRewards > 0) {
39     _depositRewards(remainingRewards, subjectProfileId);
40 }
41 return amount;
42 }
```

[GAS-03] Use errors instead of require statements as they consume less gas

Mitigation: Use errors instead of require statements as they consume less gas.

There are some errors that are not used:

```
1     error NotSupported();
2     error NotTransferable();
3     error OnlyVouchContract();
4     error OnlyVouchContract();
```

[GAS-04] Remove all occurrences of payable in EthosVouch as they are redundant

Description: `EthosVault`(`payable`(`vaultAddress`)), vault is working only with erc20's.

Mitigation: Remove all occurrences of `payable` in `EthosVouch`.